/*=======================================================================================================================
#                                                                                                                       #
#     ██████╗ ███████╗███████╗████████╗██╗███╗   ██╗██╗   ██╗████████╗███████╗███╗   ███╗██████╗ ██╗     ███████╗       #   
#     ██╔══██╗██╔════╝██╔════╝╚══██╔══╝██║████╗  ██║╚██╗ ██╔╝╚══██╔══╝██╔════╝████╗ ████║██╔══██╗██║     ██╔════╝       #
#     ██║  ██║█████╗  ███████╗   ██║   ██║██╔██╗ ██║ ╚████╔╝    ██║   █████╗  ██╔████╔██║██████╔╝██║     █████╗         #
#     ██║  ██║██╔══╝  ╚════██║   ██║   ██║██║╚██╗██║  ╚██╔╝     ██║   ██╔══╝  ██║╚██╔╝██║██╔═══╝ ██║     ██╔══╝         #
#     ██████╔╝███████╗███████║   ██║   ██║██║ ╚████║   ██║      ██║   ███████╗██║ ╚═╝ ██║██║     ███████╗███████╗       #
#     ╚═════╝ ╚══════╝╚══════╝   ╚═╝   ╚═╝╚═╝  ╚═══╝   ╚═╝      ╚═╝   ╚══════╝╚═╝     ╚═╝╚═╝     ╚══════╝╚══════╝       #
#                                                                                                                       #
=========================================================================================================================
#            __                                 __                               __                                     #
#           /  \ |  o     _  ._ _  o     _.    /  \  _  _  | _|_  _ |  _.       /  \    o _  o                          #
#          | (|/ |< | \/ (_) | | | | \/ (_|   | (|/ _> (_) |  |_ (_ | (_| \/   | (|/ >< | /_ |                          #
#           \__       /              /         \__                        /     \__                                     #
#            __                                  __                                    __                               #
#           /  \  _             _. o ._   _     /  \  _|                    _. ._     /  \  _ o  _  _.  _|  _.          #
#          | (|/ _> |_| \/ |_| (_| | | | (_|   | (|/ (_| |_| \/ |_| >< |_| (_| | |   | (|/ (_ | (_ (_| (_| (_|          #
#           \__         /        |        _|    \__          /                        \__                               #
#            __                             __                    __              __                                    #
#           /  \ o o  _. ._          _     /  \     ._ _   _     /  \     | o    /  \ |_   _. ._    |   _.              #
#          | (|/ | | (_| | | \/ |_| (/_   | (|/ |_| | | | (_)   | (|/ |_| | |   | (|/ | | (_| | |_| |< (_|              #
#           \__ _|           /             \__                   \__             \__                                    #
#                                                                                                                       #
#=======================================================================================================================*/
/**
* @custom:stories beyond stories
* 物語を超えた物語:
*
* 天语遥は実際には天语遥と呼ばれておらず、女性の名前でもありませんが、彼女が自分で選んだ女性の名前は非常に奇妙で、少し奇妙で、心地よくないので、この記事では彼女について説明します。彼女をシャオヤオと呼んでください。
* シャオヤオの実際の話は当然小説ほどドラマチックではなく、もちろん去勢前は女の子になりたかった。
* 去勢後、彼女は後悔した。
* 私が知った時は去勢された後だったのですが、知っている人から変な人だと思われていましたし、私もそう思いました。
* そして、私との出会いは彼女がネチズンに初めて会ったことで、少し光栄に思いましたが、彼女は危険で暗いオーラを発していたため、特に幸せなことではありませんでした。
* 私は彼女のことをあまり知りません、ただの噂ですが、彼女が去勢を非常に後悔しているようだということを知りました.
* だから会ったとき、あえて聞いてみた。
* 小遥 の答えはやや非論理的で、整理すると、「完全な男であろうと女であろうと、この切断された体が嫌いだ」という意味でした。
* 彼女がこの中間的な移行段階を嫌っていることを暗示しており、彼女が後悔しているのは、去勢手術の代わりに性別適合手術のために十分なお金を直接貯めるべきだということです.
* 去勢手術の理由は、当時の男性の体に耐えられなかったことと、性転換手術の費用が足りなかったということでした。お金を節約できるので、去勢手術を受けましょう。
* 私はシャオヤオのことをインターネットでよく知りませんが、彼女は私が彼女のより良い友達だと言ったので、彼女は私に会いに来ました。
* あ、ちなみに待ち合わせ場所は私の街ではなく、当時は別の街にいて、シャオヤオはその街に住んでいました。
* シャオヤオはどのように見えますか? まあ、身長は1.6メートルほどで、肌は比較的色白で、血の色はなく、そのような薄い白です。 彼女の髪は少しボサボサで、濃いくまがあり、不眠症の症状は明らかに、私が今まで見たどのメディスン ガールよりも深刻でした。
* 私は太陽を見たことのない人のように感じます。
* 彼女の容姿はなんというか、うーん…お嬢様みたい。
* またはおてんば娘。
* 見た目は十五歳くらいだが、実は十八歳。
* いくつかの点で、小遥と小晴は非常に似ています。
* 実際、正泰とロリータの差はそれほど大きくなく、少し手を加えたり、より美しい服を着たり、髪型を変えたりすれば、かわいいロリータになることもできます。 小晴行き止まりのない 360 度。
* 彼らの最大の違いは家族です。
* 小遥 は実在の子供で、家から逃げ出し、見知らぬ街で一人暮らしをしています。
* 実際、本にはXiaoqingに関する多くの物語があり、資料のソースは実際には小遥です。
* 小遥 の話はこれらよりもはるかに悪いというだけです。
* たとえば、この街に初めて来たとき、財布をなくしたり、仕事を探しているときに身分証明書を提示できなかったり、追い出されたり、殴られたりしたと彼女は言いました。
* 彼女はポケットに百元を入れて一週間暮らし、毎晩風が当たらない場所で寝ていましたが、それでもとても寒かったです。
* その時、本と同じように早春で、彼女はひどい熱を出しましたが、最後はなんとか耐えました。
* 彼女がその都市で見つけた最初の仕事は、食事と住居でしたが、給料はありませんでした。
* その後、さまざまな屈辱、困難、さらには拷問が起こりましたが、それはとにかくもっと悲劇的でした.
* 私のようなオブザーバーにとっても、多くのことが身も凍るように感じるでしょう。
* 社会の残酷さが彼女の体に凝縮されているようだった。
* 以前はそうではありませんでしたが、今は他のことをする気力があまりないようです。
* 彼女にはたくさんの悪いことが押し付けられていましたが、去勢後、それは倍増したようで、それらの悪いことはさらに多くなりました.
* 彼女はますます暗くなり始めました。
* 小慶を太陽、西子を太陽の下で努力する人、張思凡を陽と闇の間を行き来する人とすると、小姚は闇に入って出てこない人に属します。 .
* 彼女の気分は非常に不安定で、いつでも点火できる火薬樽のようです。
* 彼女の感情が突然爆発するのではないかと恐れて、私は彼女とチャットするときはいつも慎重でした.
* 普通の人が人を見るとき、彼らは人をまっすぐに見ますが、彼女は人を見るとき、頭を少し下げてから、2つの眼球を巻き上げ、一緒に眉をひそめ、白い目の広い領域を明らかにし、彼女の青白い目と相まって顔 目の下に濃いくまがある彼女の顔は、彼女を非常に恐ろしく見せました。 また、彼女は座るといつも手首を見つめていましたが、私は彼女の手首を見たことがあります。
* 彼女は何度も自傷行為をしたと言いました。
* 彼女がそう言ったとき、彼女の表情はとてもリラックスし、以前よりもさらに明るくなりました.
* 彼女いわく、機嫌が悪い時はその方法でしか落ち着かないそうです。
* 彼女にとって、自傷行為はうつ病を和らげる最良の方法です。
* しかし、普通の人の耳には、それは本当に少し神経症的なことです.
* 多くの姚娘が自傷行為をするが、それは極度に落ち込んだ時だけであり、小姚は幸福を求める方法としてこのようなものをとったのに対し、出発点はほぼ完全に異なっていると言える. 小遥 の精神に大きな問題があることは明らかです。
* 私が最も感銘を受けたのは、ウェイターが彼女に何が必要か尋ねたとき、彼はハンサムな男と美しい女の子の間でタイトルを変え、彼女を激怒させたことです.
*「私を見下してるの!?」彼女はほとんど咆哮した。
* 彼女とは別世界のような気がして、こういうシーンは映画かアニメでしか出てこないのではないかと思います。
* でも本当に目の前で起こった。
* ウェイターが謝った後、彼女の怒りはゆっくりと収まった。
* その時のレストランは、私が食べていた最も恥ずかしい時間でした. 私が食べているとき、他の人はいつも私と彼女を奇妙に見ていると感じました.
* そして、誰かがシャオヤオを長時間見ていると、彼女は突然頭を向け、その人にニヤリと笑った.
* 彼女は明らかにかなりのハンサムな女性ですが、時には野獣のように血に飢えた凶暴な行動をとります。
* これが彼女の身を守る方法なのかもしれません。
* 儚い心を隠すように、獰猛な姿で。
* 小遥の自虐傾向は非常に深刻で、いつも腕や太ももに柔らかい肉をつまんで、次々と赤い跡をつけ、見るのは痛いですが、とても快適です。
* または、自分の手を歯でかみ砕いて、深い歯の跡を残します... なぜいつもそうしているのか、私は彼女に尋ねませんでした。上。
* そして、彼女が尋ねても、彼女は説明するのが難しく、代わりにイライラするかもしれません.
* 彼女の感情的な変化については、私は本当にそれを把握することはできません.
* 私の推測によると、小遥 は、人々とコミュニケーションをとるときに、平静を保つために痛みを思い出しながら、これを行う必要があるかもしれません。
* そうしないと、彼女の感情はすぐに臨界点に達し、爆発します。
* 路上で邪魔になった炭酸飲料の缶でさえ、彼女は深い憎しみを持っているかのように歯を食いしばり、彼女を激しく蹴ります.
* 精神科に受診を勧めたかったのですが、面識がなく、怒られるのではないかと心に秘めていました。トラブルに巻き込まれることをより恐れています。
* 小遥とはあまりコミュニケーションをとらず、ほとんど無言で、行った場所はレストランやショッピングモールで、別れました。
* コミュニケーションは少なかったのですが、精神病患者のような感覚がとても印象的でした。
* 彼女はおそらく軽度または中等度の精神疾患のレベルに達していると思います。
* ちなみに、彼女の自傷行為は本当に深刻です。 会っている間は自分をつまんだり噛んだり、離れるときは自分の指を噛んだり、指から血が滴っているのがはっきりと見えましたが、とても簡単に起き上がるように見えました。
* 彼女の体に見られるすべての傷を考えると、寒気がしました。
* 私は彼女の経験に同情しますが、私はまだ彼女に近づきたくありません.
* 彼女はあまりにも危険を感じているので、いつの日か完全に爆発し、ナイフで何人かをハッキングして死に至らしめる可能性があります。
* 夕子に関する黒くなった内容は、実際の小遥の将来についての私の想像から来ています.小遥の家族の背景についてはあまり尋ねませんでした。お気に入り。
* シャオヤオの家族環境が非常に悪いことだけは知っています.それは田舎の家族であり,彼女の両親は娘よりも息子を好みます.彼女の家族には姉と兄と弟がいます.人口は非常に多く,5人いると言われています. 小遥 を含む人々彼は子供なので、小遥 が家出することをあまり気にしていないようです。
* 彼らは小遥のことを知っていて、さらに... 小遥によると、彼らは小遥を追い払った。
* 荷物を投げ出して、この家から出るように言ってください。
* 都会育ちの私には想像もつかないことです。
* でも辺鄙な村なら、こんなこともあり得ないわけではない……そんな共感があったからか、小姚の両親を小説の中で共感的な両親に設定したところ、家族はもう貧しくなく、でも裕福な家庭。 また、小遥 の能動的去勢を受動的去勢に変更しました... ストーリーはまだ良くありませんが、実際にはかなり良いです。
* 多くの視聴者は本の章が残酷すぎると感じていますが、実際には、現実に苦しんでいる多くの麻薬少女にとって、本の世界もとても美しいです.
* 現実はフィクションよりも残酷であり、ほとんどの読者の気持ちを考慮したため、いくつかの美化を行いました.
* シャオヤオの最終的な精神疾患については、実際の彼女と似ていて、彼女の精神状態がこのまま続くと、遅かれ早かれ間違いなく統合失調症になると思います。
* それで、小遥の実際の話については、それだけです。
*
*「メディシンガールの空」より抜粋.
*/
// File: @openzeppelin/contracts/utils/StorageSlot.sol
// OpenZeppelin Contracts (last updated v4.7.0) (utils/StorageSlot.sol)
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.10;
/**
 * @dev Library for reading and writing primitive types to specific storage slots.
 *
 * Storage slots are often used to avoid storage conflict when dealing with upgradeable contracts.
 * This library helps with reading and writing to such slots without the need for inline assembly.
 *
 * The functions in this library return Slot structs that contain a `value` member that can be used to read or write.
 *
 * Example usage to set ERC1967 implementation slot:
 * ```
 * contract ERC1967 {
 *     bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;
 *
 *     function _getImplementation() internal view returns (address) {
 *         return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
 *     }
 *
 *     function _setImplementation(address newImplementation) internal {
 *         require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
 *         StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
 *     }
 * }
 * ```
 *
 * _Available since v4.1 for `address`, `bool`, `bytes32`, and `uint256`._
 */
library StorageSlot {
    struct AddressSlot {
        address value;
    }
    struct BooleanSlot {
        bool value;
    }
    struct Bytes32Slot {
        bytes32 value;
    }
    struct Uint256Slot {
        uint256 value;
    }
    /**
     * @dev Returns an `AddressSlot` with member `value` located at `slot`.
     */
    function getAddressSlot(bytes32 slot) internal pure returns (AddressSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
    /**
     * @dev Returns an `BooleanSlot` with member `value` located at `slot`.
     */
    function getBooleanSlot(bytes32 slot) internal pure returns (BooleanSlot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
    /**
     * @dev Returns an `Bytes32Slot` with member `value` located at `slot`.
     */
    function getBytes32Slot(bytes32 slot) internal pure returns (Bytes32Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
    /**
     * @dev Returns an `Uint256Slot` with member `value` located at `slot`.
     */
    function getUint256Slot(bytes32 slot) internal pure returns (Uint256Slot storage r) {
        /// @solidity memory-safe-assembly
        assembly {
            r.slot := slot
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)
pragma solidity ^0.8.10;
/**
 * @dev Collection of functions related to the address type
 */
library Address {

    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }
    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }
    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }
    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }
    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/interfaces/draft-IERC1822.sol
// OpenZeppelin Contracts (last updated v4.5.0) (interfaces/draft-IERC1822.sol)
pragma solidity ^0.8.10;
/**
 * @dev ERC1822: Universal Upgradeable Proxy Standard (UUPS) documents a method for upgradeability through a simplified
 * proxy whose upgrades are fully controlled by the current implementation.
 */
interface IERC1822Proxiable {
    /**
     * @dev Returns the storage slot that the proxiable contract assumes is being used to store the implementation
     * address.
     *
     * IMPORTANT: A proxy pointing at a proxiable contract should not be considered proxiable itself, because this risks
     * bricking a proxy that upgrades to it, by delegating to itself until out of gas. Thus it is critical that this
     * function revert if invoked through a proxy.
     */
    function proxiableUUID() external view returns (bytes32);
}

// File: @openzeppelin/contracts/proxy/beacon/IBeacon.sol
// OpenZeppelin Contracts v4.4.1 (proxy/beacon/IBeacon.sol)
pragma solidity ^0.8.10;
/**
 * @dev This is the interface that {BeaconProxy} expects of its beacon.
 */
interface IBeacon {
    /**
     * @dev Must return an address that can be used as a delegate call target.
     *
     * {BeaconProxy} will check that this address is a contract.
     */
    function implementation() external view returns (address);
}

// File: @openzeppelin/contracts/proxy/ERC1967/ERC1967Upgrade.sol
// OpenZeppelin Contracts (last updated v4.5.0) (proxy/ERC1967/ERC1967Upgrade.sol)
pragma solidity ^0.8.10;
/**
 * @dev This abstract contract provides getters and event emitting update functions for
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967] slots.
 *
 * _Available since v4.1._
 *
 * @custom:oz-upgrades-unsafe-allow delegatecall
 */
abstract contract ERC1967Upgrade {
    // This is the keccak-256 hash of "eip1967.proxy.rollback" subtracted by 1
    bytes32 private constant _ROLLBACK_SLOT = 0x4910fdfa16fed3260ed0e7147f7cc6da11a60208b5b9406d12a635614ffd9143;

    /**
     * @dev Storage slot with the address of the current implementation.
     * This is the keccak-256 hash of "eip1967.proxy.implementation" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _IMPLEMENTATION_SLOT = 0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc;

    /**
     * @dev Emitted when the implementation is upgraded.
     */
    event Upgraded(address indexed implementation);
    /**
     * @dev Returns the current implementation address.
     */
    function _getImplementation() internal view returns (address) {
        return StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value;
    }
    /**
     * @dev Stores a new address in the EIP1967 implementation slot.
     */
    function _setImplementation(address newImplementation) private {
        require(Address.isContract(newImplementation), "ERC1967: new implementation is not a contract");
        StorageSlot.getAddressSlot(_IMPLEMENTATION_SLOT).value = newImplementation;
    }
    /**
     * @dev Perform implementation upgrade
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeTo(address newImplementation) internal {
        _setImplementation(newImplementation);
        emit Upgraded(newImplementation);
    }
    /**
     * @dev Perform implementation upgrade with additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCall(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        _upgradeTo(newImplementation);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(newImplementation, data);
        }
    }
    /**
     * @dev Perform implementation upgrade with security checks for UUPS proxies, and additional setup call.
     *
     * Emits an {Upgraded} event.
     */
    function _upgradeToAndCallUUPS(
        address newImplementation,
        bytes memory data,
        bool forceCall
    ) internal {
        // Upgrades from old implementations will perform a rollback test. This test requires the new
        // implementation to upgrade back to the old, non-ERC1822 compliant, implementation. Removing
        // this special case will break upgrade paths from old UUPS implementation to new ones.
        if (StorageSlot.getBooleanSlot(_ROLLBACK_SLOT).value) {
            _setImplementation(newImplementation);
        } else {
            try IERC1822Proxiable(newImplementation).proxiableUUID() returns (bytes32 slot) {
                require(slot == _IMPLEMENTATION_SLOT, "ERC1967Upgrade: unsupported proxiableUUID");
            } catch {
                revert("ERC1967Upgrade: new implementation is not UUPS");
            }
            _upgradeToAndCall(newImplementation, data, forceCall);
        }
    }
    /**
     * @dev Storage slot with the admin of the contract.
     * This is the keccak-256 hash of "eip1967.proxy.admin" subtracted by 1, and is
     * validated in the constructor.
     */
    bytes32 internal constant _ADMIN_SLOT = 0xb53127684a568b3173ae13b9f8a6016e243e63b6e8ee1178d6a717850b5d6103;
    /**
     * @dev Emitted when the admin account has changed.
     */
    event AdminChanged(address previousAdmin, address newAdmin);
    /**
     * @dev Returns the current admin.
     */
    function _getAdmin() internal view returns (address) {
        return StorageSlot.getAddressSlot(_ADMIN_SLOT).value;
    }
    /**
     * @dev Stores a new address in the EIP1967 admin slot.
     */
    function _setAdmin(address newAdmin) private {
        require(newAdmin != address(0), "ERC1967: new admin is the zero address");
        StorageSlot.getAddressSlot(_ADMIN_SLOT).value = newAdmin;
    }
    /**
     * @dev Changes the admin of the proxy.
     *
     * Emits an {AdminChanged} event.
     */
    function _changeAdmin(address newAdmin) internal {
        emit AdminChanged(_getAdmin(), newAdmin);
        _setAdmin(newAdmin);
    }

    /**
     * @dev The storage slot of the UpgradeableBeacon contract which defines the implementation for this proxy.
     * This is bytes32(uint256(keccak256('eip1967.proxy.beacon')) - 1)) and is validated in the constructor.
     */
    bytes32 internal constant _BEACON_SLOT = 0xa3f0ad74e5423aebfd80d3ef4346578335a9a72aeaee59ff6cb3582b35133d50;

    /**
     * @dev Emitted when the beacon is upgraded.
     */
    event BeaconUpgraded(address indexed beacon);

    /**
     * @dev Returns the current beacon.
     */
    function _getBeacon() internal view returns (address) {
        return StorageSlot.getAddressSlot(_BEACON_SLOT).value;
    }
    /**
     * @dev Stores a new beacon in the EIP1967 beacon slot.
     */
    function _setBeacon(address newBeacon) private {
        require(Address.isContract(newBeacon), "ERC1967: new beacon is not a contract");
        require(
            Address.isContract(IBeacon(newBeacon).implementation()),
            "ERC1967: beacon implementation is not a contract"
        );
        StorageSlot.getAddressSlot(_BEACON_SLOT).value = newBeacon;
    }
    /**
     * @dev Perform beacon upgrade with additional setup call. Note: This upgrades the address of the beacon, it does
     * not upgrade the implementation contained in the beacon (see {UpgradeableBeacon-_setImplementation} for that).
     *
     * Emits a {BeaconUpgraded} event.
     */
    function _upgradeBeaconToAndCall(
        address newBeacon,
        bytes memory data,
        bool forceCall
    ) internal {
        _setBeacon(newBeacon);
        emit BeaconUpgraded(newBeacon);
        if (data.length > 0 || forceCall) {
            Address.functionDelegateCall(IBeacon(newBeacon).implementation(), data);
        }
    }
}

// File: @openzeppelin/contracts/proxy/Proxy.sol
// OpenZeppelin Contracts (last updated v4.6.0) (proxy/Proxy.sol)
pragma solidity ^0.8.10;
/**
 * @dev This abstract contract provides a fallback function that delegates all calls to another contract using the EVM
 * instruction `delegatecall`. We refer to the second contract as the _implementation_ behind the proxy, and it has to
 * be specified by overriding the virtual {_implementation} function.
 *
 * Additionally, delegation to the implementation can be triggered manually through the {_fallback} function, or to a
 * different contract through the {_delegate} function.
 *
 * The success and return data of the delegated call will be returned back to the caller of the proxy.
 */
abstract contract Proxy {
    /**
     * @dev Delegates the current call to `implementation`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _delegate(address implementation) internal virtual {
        assembly {
            // Copy msg.data. We take full control of memory in this inline assembly
            // block because it will not return to Solidity code. We overwrite the
            // Solidity scratch pad at memory position 0.
            calldatacopy(0, 0, calldatasize())

            // Call the implementation.
            // out and outsize are 0 because we don't know the size yet.
            let result := delegatecall(gas(), implementation, 0, calldatasize(), 0, 0)

            // Copy the returned data.
            returndatacopy(0, 0, returndatasize())

            switch result
            // delegatecall returns 0 on error.
            case 0 {
                revert(0, returndatasize())
            }
            default {
                return(0, returndatasize())
            }
        }
    }

    /**
     * @dev This is a virtual function that should be overridden so it returns the address to which the fallback function
     * and {_fallback} should delegate.
     */
    function _implementation() internal view virtual returns (address);
    
    /**
     * @dev Delegates the current call to the address returned by `_implementation()`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _fallback() internal virtual {
        _beforeFallback();
        _delegate(_implementation());
    }
    
    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if no other
     * function in the contract matches the call data.
     */
    fallback() external payable virtual {
        _fallback();
    }
    
    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if call data
     * is empty.
     */
    receive() external payable virtual {
        _fallback();
    }
    /**
     * @dev Hook that is called before falling back to the implementation. Can happen as part of a manual `_fallback`
     * call, or as part of the Solidity `fallback` or `receive` functions.
     *
     * If overridden should call `super._beforeFallback()`.
     */
    function _beforeFallback() internal virtual {}
}

// File: @openzeppelin/contracts/proxy/ERC1967/ERC1967Proxy.sol
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/ERC1967/ERC1967Proxy.sol)
pragma solidity ^0.8.10;
/**
 * @dev This contract implements an upgradeable proxy. It is upgradeable because calls are delegated to an
 * implementation address that can be changed. This address is stored in storage in the location specified by
 * https://eips.ethereum.org/EIPS/eip-1967[EIP1967], so that it doesn't conflict with the storage layout of the
 * implementation behind the proxy.
 */
contract ERC1967Proxy is Proxy, ERC1967Upgrade {
    /**
     * @dev Initializes the upgradeable proxy with an initial implementation specified by `_logic`.
     *
     * If `_data` is nonempty, it's used as data in a delegate call to `_logic`. This will typically be an encodePackedd
     * function call, and allows initializing the storage of the proxy like a Solidity constructor.
     */
    constructor(address _logic, bytes memory _data) payable {
        _upgradeToAndCall(_logic, _data, false);
    }
    /**
     * @dev Returns the current implementation address.
     */
    function _implementation() internal view virtual override returns (address impl) {
        return ERC1967Upgrade._getImplementation();
    }
}

// File: @openzeppelin/contracts/proxy/transparent/TransparentUpgradeableProxy.sol
// OpenZeppelin Contracts (last updated v4.7.0) (proxy/transparent/TransparentUpgradeableProxy.sol)
pragma solidity ^0.8.10;
/**
 * @dev This contract implements a proxy that is upgradeable by an admin.
 *
 * To avoid https://medium.com/nomic-labs-blog/malicious-backdoors-in-ethereum-proxies-62629adf3357[proxy selector
 * clashing], which can potentially be used in an attack, this contract uses the
 * https://blog.openzeppelin.com/the-transparent-proxy-pattern/[transparent proxy pattern]. This pattern implies two
 * things that go hand in hand:
 *
 * 1. If any account other than the admin calls the proxy, the call will be forwarded to the implementation, even if
 * that call matches one of the admin functions exposed by the proxy itself.
 * 2. If the admin calls the proxy, it can access the admin functions, but its calls will never be forwarded to the
 * implementation. If the admin tries to call a function on the implementation it will fail with an error that says
 * "admin cannot fallback to proxy target".
 *
 * These properties mean that the admin account can only be used for admin actions like upgrading the proxy or changing
 * the admin, so it's best if it's a dedicated account that is not used for anything else. This will avoid headaches due
 * to sudden errors when trying to call a function from the proxy implementation.
 *
 * Our recommendation is for the dedicated account to be an instance of the {ProxyAdmin} contract. If set up this way,
 * you should think of the `ProxyAdmin` instance as the real administrative interface of your proxy.
 */
contract TransparentUpgradeableProxy is ERC1967Proxy {
    /**
     * @dev Initializes an upgradeable proxy managed by `_admin`, backed by the implementation at `_logic`, and
     * optionally initialized with `_data` as explained in {ERC1967Proxy-constructor}.
     */
    constructor(
        address _logic,
        address admin_,
        bytes memory _data
    ) payable ERC1967Proxy(_logic, _data) {
        _changeAdmin(admin_);
    }
    /**
     * @dev Modifier used internally that will delegate the call to the implementation unless the sender is the admin.
     */
    modifier ifAdmin() {
        if (msg.sender == _getAdmin()) {
            _;
        } else {
            _fallback();
        }
    }

    /**
     * @dev Returns the current admin.
     *
     * NOTE: Only the admin can call this function. See {ProxyAdmin-getProxyAdmin}.
     *
     * TIP: To get this value clients can read directly from the storage slot shown below (specified by EIP1967) using the
     * https://eth.wiki/json-rpc/API#eth_getstorageat[`eth_getStorageAt`] RPC call.
     * `0xb53127684a568b3173ae13b9f8a6016e243e63b6e8ee1178d6a717850b5d6103`
     */
    function admin() external ifAdmin returns (address admin_) {
        admin_ = _getAdmin();
    }

    /**
     * @dev Returns the current implementation.
     *
     * NOTE: Only the admin can call this function. See {ProxyAdmin-getProxyImplementation}.
     *
     * TIP: To get this value clients can read directly from the storage slot shown below (specified by EIP1967) using the
     * https://eth.wiki/json-rpc/API#eth_getstorageat[`eth_getStorageAt`] RPC call.
     * `0x360894a13ba1a3210667c828492db98dca3e2076cc3735a920a3ca505d382bbc`
     */
    function implementation() external ifAdmin returns (address implementation_) {
        implementation_ = _implementation();
    }
    
    /**
     * @dev Changes the admin of the proxy.
     *
     * Emits an {AdminChanged} event.
     *
     * NOTE: Only the admin can call this function. See {ProxyAdmin-changeProxyAdmin}.
     */
    function changeAdmin(address newAdmin) external virtual ifAdmin {
        _changeAdmin(newAdmin);
    }
    /**
     * @dev Upgrade the implementation of the proxy.
     *
     * NOTE: Only the admin can call this function. See {ProxyAdmin-upgrade}.
     */
    function upgradeTo(address newImplementation) external ifAdmin {
        _upgradeToAndCall(newImplementation, bytes(""), false);
    }
    /**
     * @dev Upgrade the implementation of the proxy, and then call a function from the new implementation as specified
     * by `data`, which should be an encodePackedd function call. This is useful to initialize new storage variables in the
     * proxied contract.
     *
     * NOTE: Only the admin can call this function. See {ProxyAdmin-upgradeAndCall}.
     */
    function upgradeToAndCall(address newImplementation, bytes calldata data) external payable ifAdmin {
        _upgradeToAndCall(newImplementation, data, true);
    }
    /**
     * @dev Returns the current admin.
     */
    function _admin() internal view virtual returns (address) {
        return _getAdmin();
    }
    /**
     * @dev Makes sure the admin cannot access the fallback function. See {Proxy-_beforeFallback}.
     */
    function _beforeFallback() internal virtual override {
        require(msg.sender != _getAdmin(), "TransparentUpgradeableProxy: admin cannot fallback to proxy target");
        super._beforeFallback();
    }
}

// File: @openzeppelin/contracts/proxy/transparent/ProxyAdmin.sol
// OpenZeppelin Contracts v4.4.1 (proxy/transparent/ProxyAdmin.sol)
pragma solidity ^0.8.10;
/**
 * @dev This is an auxiliary contract meant to be assigned as the admin of a {TransparentUpgradeableProxy}. For an
 * explanation of why you would want to use this see the documentation for {TransparentUpgradeableProxy}.
 */
contract IProxyAdmin {
    address private _proxyAdminOwner;
    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_proxyAdminOwner == msg.sender, "Ownable: caller is not the owner");
        _;
    }
    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    function IProxyAdmin_Initialize() internal {
        _proxyAdminOwner = address(this);
    }

    /**
     * @dev Returns the address of the current owner.
     * @notice Returns the address of the current owner.
     */
    function _getProxyAdminOwner() public view virtual returns (address) {
        return _proxyAdminOwner;
    }
    /**
     * @notice Returns the current implementation of `proxy`.
     * @dev Returns the current implementation of `proxy`.
     *
     * Requirements:
     *
     * - This contract must be the admin of `proxy`.
     */
    function _getProxyImplementation(TransparentUpgradeableProxy proxy) public view virtual returns (address) {
        // We need to manually run the static call since the getter cannot be flagged as view
        // bytes4(keccak256("implementation()")) == 0x5c60da1b
        (bool success, bytes memory returndata) = address(proxy).staticcall(hex"5c60da1b");
        require(success);
        return abi.decode(returndata, (address));
    }
    
    /**
     * @notice Returns the current admin of `proxy`.
     * @dev Returns the current admin of `proxy`.
     *
     * Requirements:
     *
     * - This contract must be the admin of `proxy`.
     */
    function _getProxyAdmin(TransparentUpgradeableProxy proxy) public view virtual returns (address) {
        // We need to manually run the static call since the getter cannot be flagged as view
        // bytes4(keccak256("admin()")) == 0xf851a440
        (bool success, bytes memory returndata) = address(proxy).staticcall(hex"f851a440");
        require(success);
        return abi.decode(returndata, (address));
    }
    
    /**
     * @notice Changes the admin of `proxy` to `newAdmin`.
     * @dev Changes the admin of `proxy` to `newAdmin`.
     *
     * Requirements:
     *
     * - This contract must be the current admin of `proxy`.
     */
    function changeProxyAdmin(TransparentUpgradeableProxy proxy, address newAdmin) public virtual onlyOwner {
        proxy.changeAdmin(newAdmin);
    }
    
    /**
     * @notice Upgrades `proxy` to `implementation`. See {TransparentUpgradeableProxy-upgradeTo}.
     * @dev Upgrades `proxy` to `implementation`. See {TransparentUpgradeableProxy-upgradeTo}.
     *
     * Requirements:
     *
     * - This contract must be the admin of `proxy`.
     */
    function upgrade(TransparentUpgradeableProxy proxy, address implementation) public virtual onlyOwner {
        proxy.upgradeTo(implementation);
    }
    
    /**
     * @dev Upgrades `proxy` to `implementation` and calls a function on the new implementation. See
     * {TransparentUpgradeableProxy-upgradeToAndCall}.
     * @notice Upgrades `proxy` to `implementation` and calls a function on the new implementation. See
     * {TransparentUpgradeableProxy-upgradeToAndCall}.
     *
     * Requirements:
     *
     * - This contract must be the admin of `proxy`.
     */
    function upgradeAndCall(
        TransparentUpgradeableProxy proxy,
        address implementation,
        bytes memory data
    ) public payable virtual onlyOwner {
        proxy.upgradeToAndCall{value: msg.value}(implementation, data);
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)
pragma solidity ^0.8.10;
/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155Receiver.sol
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC1155/IERC1155Receiver.sol)
pragma solidity ^0.8.10;
/**
 * @dev _Available since v3.1._
 */
interface IERC1155Receiver is IERC165 {
    
    /**
     * @dev Handles the receipt of a single ERC1155 token type. This function is
     * called at the end of a `safeTransferFrom` after the balance has been updated.
     *
     * NOTE: To accept the transfer, this must return
     * `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))`
     * (i.e. 0xf23a6e61, or its own function selector).
     *
     * @param operator The address which initiated the transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param id The ID of the token being transferred
     * @param value The amount of tokens being transferred
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))` if transfer is allowed
     */
    function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes calldata data
    ) external returns (bytes4);

    /**
     * @dev Handles the receipt of a multiple ERC1155 token types. This function
     * is called at the end of a `safeBatchTransferFrom` after the balances have
     * been updated.
     *
     * NOTE: To accept the transfer(s), this must return
     * `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))`
     * (i.e. 0xbc197c81, or its own function selector).
     *
     * @param operator The address which initiated the batch transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param ids An array containing ids of each token being transferred (order and length must match values array)
     * @param values An array containing amounts of each token being transferred (order and length must match ids array)
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))` if transfer is allowed
     */
    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] calldata ids,
        uint256[] calldata values,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)
pragma solidity ^0.8.10;
/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to approve the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)
pragma solidity ^0.8.10;
/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);
    
    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
    
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);
    
    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);
    
    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);
    
    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);
    
    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);
    
    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}
/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/

// File: contracts/ITweet.sol
pragma solidity ^0.8.10;
/// @notice Integrated, that allows anyone to tweet.
contract ITweet{
    event TweetIssued(uint indexed tweetID, string indexed tweet_String);

    struct Tweet {
		uint tweetID;
		string tweet;
	}

    Tweet[] internal _tweets;
    bytes constant internal WORLD_LINE_START = "0xe5a4a9e7a9bae69c89e4ba9be998b4e6b289e6b289e79a842ce6b585e781b0e889b2e79a84e4ba91e981aee68ca1e79d80e5a4a9e7a9ba2ce4b99fe981aee68ca1e79d80e982a3e69cace5ba94e8afa5e58d81e58886e88080e79cbce79a84e5a4aae998b32e0a0a54686520736b792077617320612062697420676c6f6f6d792c2077697468206c69676874206772617920636c6f75647320636f766572696e672074686520736b7920616e64207468652073756e20746861742073686f756c642068617665206265656e2064617a7a6c696e672e";
    bytes constant internal WORLD_LINE_END = "0xe5a4a9e7a9bae698afe982a3e6a0b7e79a84e6b99be8939d2ce6b99be8939de79a84e5a5bde5838fe5aeb9e4b88de4b88be4b880e4b89de585b6e4bb96e79a84e889b2e5bda92e0a0a54686520736b7920697320736f20626c75652c2074686520626c7565207365656d7320746f20626520756e61626c6520746f206163636f6d6d6f646174652061207472616365206f66206f7468657220636f6c6f72732e";

    constructor(){
        _tweets.push(
            Tweet({
                tweetID: _tweets.length,
                tweet: string(WORLD_LINE_START)
            })
        );
    }
    /// @notice Allow everyone to post uncensored tweets, and the content will be permanently stored on the blockchain.
    function IssueTweet(string memory tweet)
	public 
	returns (uint _tweetID)
	{
		_tweets.push(
			Tweet({
				tweetID : _tweets.length,
				tweet : tweet
			})
		);
		emit TweetIssued(_tweets.length, tweet);
		return _tweets.length;
	}
    function _getTweet(uint _tweetID) public view returns (Tweet memory){
		return _tweets[_tweetID];
	}
    function _getLastTweetID() public view returns (uint) {
		return _tweets.length - 1;
	}
}

// File: contracts/IDIP.sol
pragma solidity ^0.8.10;
/// @notice Integrated, Used to record Destiny Improvement Proposal (DIP) and other important matters, only the owner can publish.
contract IDIP{
	event DIPIssued(uint indexed DIP_ID);
	
	struct DIP {
		uint DIP_ID;
		string data;
	}

	DIP[] internal _DIPs;
    bytes constant internal STORY_START = "0xe28094e28094e4b88de684bfe6848fe689bfe8aea4e887aae5b7b1e698afe794b7e5ada9e5ad90e79a84e4ba8be5ae9e2ce8808ce4bdbfe794a8e88dafe789a9e8aea9e887aae5b7b1e58f98e5be97e69bb4e5838fe5a5b3e5ada9e5ad90e4b880e4ba9be79a84e5a5b9e4bbac2ce4bcbce4b98ee698afe59ca8e69c9de79d80e6a2a6e683b3e5898de8bf9b2ce58db4e58f88e69bb4e5838fe698afe59ca8e887aae68891e9babbe986892e0ae6a2a62ce7bb88e5b086e4bc9ae986922ce982a3e781b0e889b2e8808ce69ca6e883a7e79a84e5a4a9e7a9ba2ce4b99fe7bb88e698afe8a681e99da2e5afb92e2e2e2e2e2e0a0ae28094e280944920646f6ee28099742077616e7420746f2061646d69742074686520666163742074686174204920616d206120626f792c2062757420746865792075736520647275677320746f206d616b65207468656d73656c766573206d6f7265206c696b65206769726c732c206974207365656d732074686174207468657920617265206d6f76696e6720746f776172647320746865697220647265616d732c20627574206974206973206d6f7265206c696b652073656c662d616e65737468657369612e0a447265616d732077696c6c206576656e7475616c6c792077616b652075702c20616e6420746865206772617920616e642068617a7920736b792077696c6c206576656e7475616c6c7920666163652e2e2e";
    bytes constant internal STORY_END = "0xe7949fe5898de4b88de883bde698afe4b8aae5a5b3e5ada9e5ad902ce982a3e4b9882ce5b0b1e8aea9e68891e4bba5e5a5b3e5ada9e79a84e6a8a1e6a0b7e6adbbe58ebbe590a72e0a0ae5868de8a7812e0a0a4920636f756c646e27742062652061206769726c207768656e20492077617320616c6976652c20736f206c6574206d65206469652061732061206769726c2e0a0a6279652e";

	constructor(){
        _DIPs.push(
            DIP({
                DIP_ID: _DIPs.length,
                data: string(STORY_START)
            })
        );
    }

	function _getLastDIP_ID() public view returns (uint) {
		return _DIPs.length - 1;
	}
	function _getDIP(uint _DIP_ID) public view returns(DIP memory){
		return _DIPs[_DIP_ID];
	}
}

// File: contracts/ICreate2Factory.sol
pragma solidity ^0.8.10;
/// @notice Integrated, used to deploy contracts using Create2, only the owner can call.
contract IDestinyDeployer{
    event Deployed(address addr, bytes32 salt);
    /// @notice Calculate create2 deploy contract address.
    function getAddress(bytes memory bytecode, string memory salt)
        public
        view
        returns (address)
    {
        bytes32 hash = keccak256(
            abi.encodePacked(bytes1(0xff), address(this), keccak256(abi.encodePacked(salt)), keccak256(bytecode))
        );

        // NOTE: cast last 20 bytes of hash to address
        return address(uint160(uint(hash)));
    }
    
    /**
    * @custom:title Deploy the contract
    * @notice Check the event log Deployed which contains the address of the deployed Contract.
    * The address in the log should equal the address computed from above.
    */
    function deploy(bytes memory bytecode, bytes32 salt) internal returns(bool){
        address addr;
        /*
        NOTE: How to call create2

        create2(v, p, n, s)
        create new contract with code at memory p to p + n
        and send v wei
        and return the new address
        where new address = first 20 bytes of keccak256(0xff + address(this) + s + keccak256(mem[p…(p+n)))
              s = big-endian 256-bit value
        */
        assembly {
            addr := create2(
                callvalue(), // wei sent with current call
                // Actual code starts after skipping the first 32 bytes
                add(bytecode, 0x20),
                mload(bytecode), // Load the size of code contained in the first 32 bytes
                salt // Salt from function arguments
            )
            if iszero(extcodesize(addr)) {
                revert(0, 0)
            }
        }
        emit Deployed(addr, salt);
        return true;
    }
}
/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/

// File: contracts/IMultiSigAdmin.sol
pragma solidity ^0.8.10;
/// @notice Used to manage multi-signature contracts.
contract IMultiSigAdmin{
	event OwnerAddition(address indexed owner);
	event OwnerRemoval(address indexed owner);
	event SignatureModify(string signature);
	event MinApproveCountModify(uint minApproveCount);
	event TransactionLockModify(uint transactionLockTime);
	event ContractDisabled(uint enableTime);
	event ContractEnabled(bool success, uint enableTime);
	event SelfDestruction(address ultimateBeneficiaryAddress);

    bool internal initialized;
	uint constant internal MAX_OWNER_COUNT = 49;
	address[] internal owners;
	mapping(address => bool) internal isOwner;
	uint internal minApproveCount = 5;
	bool internal contractDisabled;
	uint internal transactionLockTime = 86400;
    ///@dev initial owners.
    //Haruka-story recorder,generally does not participate in approval.
    address constant internal KIYOMIYA = 0x00001C1D6ab92F943eD4A31dA8F447Fd96589960;
    address constant internal SOFTCLAY = 0x11119C3A27d5D7E13cb52053aF58b2DBddcFE051;
    address constant internal XIZI = 0x22222eC77C520Bdb7D6A2450C3dB3c5c138C4372;
    address constant internal SUYUQING = 0x33339BE5D3C5C7ae99c1532df8a09F859770B3E3;
    address constant internal DUYUXUAN = 0x4444023B8E794eCD3a21335fcA22675739bD7914;
    address constant internal CICADA = 0x555599F812DC2Cf428d67339221e2B066e7fCAe5;
    address constant internal JIANYUE = 0x66660Bd655e77b2d8b0Ad9F87b4c48D7f284E9b6;
    address constant internal UMO = 0x77777DCaEfeaC067f21162cd2F48E5b5dB0A2B97;
    address constant internal ULI = 0x888853CFdAB45eB0608Acc157C6295E8eFD617a8;
    address constant internal HARUKA = 0x99995D080A1bfa91d065dD14C567089D103BfBB9;

    constructor(){
        address[10] memory excutors = [KIYOMIYA,SOFTCLAY,XIZI,SUYUQING,DUYUXUAN,CICADA,JIANYUE,UMO,ULI,HARUKA];
        for(uint i=0;i<excutors.length;i++){
            owners.push(excutors[i]);
            isOwner[excutors[i]] = true;
        }
    }

    modifier onlyDESTINY_EXCUTORS(){
        require(msg.sender == KIYOMIYA || msg.sender == SOFTCLAY || msg.sender == XIZI || msg.sender == SUYUQING || msg.sender == DUYUXUAN || msg.sender == CICADA ||msg.sender == JIANYUE || msg.sender == UMO || msg.sender == ULI || msg.sender == HARUKA,"NOT DESTINY EXCUTORS.");
        _;    
    }
    modifier verifyInitialize(){
        require(!initialized,"Initialized.");
        initialized = true;
        _;
    }
    modifier onlyOwners() {
		require(isOwner[msg.sender], "Msg.sender not owner.");
		_;
	}
	modifier ownerDoesNotExist(address owner) {
		require(!isOwner[owner], "Owner already exist.");
		_;
	}
	modifier ownerExists(address owner) {
		require(isOwner[owner], "Owner does not exist.");
		_;
	}
	modifier verifyOwner(address owner) {
		require(owner != address(0), "Owner cannot be zero address");
		require(owner != address(this), "Owner cannot be this contract address.");
		_;
	}
	modifier verifyMinApproveCount(uint _minApproveCount,uint ownerCount) {
		require(
			ownerCount <= MAX_OWNER_COUNT
			&& _minApproveCount <= ownerCount
			&& _minApproveCount > 0
			&& ownerCount != 0,
			"MinApproveCount not valid."
		);
		_;
	}
	modifier verifyContractStauts(uint _transactionType){
		if (_transactionType != 9) {
			require(!contractDisabled, "Contract already disabled.");
		}
		_;
	}
	modifier verifyTransactionLockTime(uint _transactionLockTime){
		require(
			_transactionLockTime >= 86400
			&& _transactionLockTime <= 6998400,
			"TransactionLockTime must be greater than 86400 seconds and must be less than 6998400 seconds."
		);
		_;
	}
    function _getDESTINY_EXCUTORS() public pure  returns (address[10] memory DESTINY_EXCUTORS){
        return([KIYOMIYA,SOFTCLAY,XIZI,SUYUQING,DUYUXUAN,CICADA,JIANYUE,UMO,ULI,HARUKA]);
    }
	function getMAX_OWNER_COUNT() public pure  returns (uint){
		return MAX_OWNER_COUNT;
	}
	function _getMultiSigOwners() public view returns (address[] memory) {
		return owners;
	}
	function getIsMultiSigOwner(address owner) public view returns (bool){
		return isOwner[owner];
	}
	function _getMinApproveCount() public view returns (uint){
		return minApproveCount;
	}
	function _getContractDisabled() public view returns (bool){
		return contractDisabled;
	}
	function _getTransactionLockTime() public view returns (uint){
		return transactionLockTime;
	}
}

// File: contracts/IMultiSigTransaction.sol
pragma solidity ^0.8.10;
/// @notice Used to maintain multi-signature contract __transactions.
contract IMultiSigTransaction{
	event Deposit(address indexed sender, uint amount, uint balance);
	event IssueTransaction
	(
		address indexed msgSender,
		uint indexed transactionID,
		address indexed to,
		Transaction _transaction
	);
	event ApproveTransaction(address indexed msgSender, uint indexed transactionID);
	event RevokeApproveaction(address indexed msgSender, uint indexed transactionID);
	event ExecuteTransaction(address indexed msgSender, uint indexed transactionID);
		
	struct Transaction {
		uint transactionID;
		address to;
		uint value;
		bytes data;
		string transactionTypeStr;
		uint transactionType;
		uint approveCount;
		uint canExecuteTimeStamp;
		bool executed;
		string message;
		string transactionSignature;
	}

	mapping(uint => mapping(address => bool)) internal isApproved;
	uint[12] internal transactionTypes=[0,1,2,3,4,5,6,7,8,9,10,11];
	string[12]  internal transactionTypeStrs=["Transaction",
		"IssueDIP",
		"ContractDeploy",
		"AddOwner",
		"RemoveOwner",
		"ModifySignature",
		"ModifyMinApproveCount",
		"ModifyTransactionLock",
		"DisableTransactionLock",
		"EnableContract",
		"DisableContract",
		"SelfDestruct"
	];
	Transaction[] public  __transactions;
	string internal signature = "destinytemple.eth";
	bool internal reentrancyLocked;

	constructor(){
	 	uint _transactionType = 9;
	 	string memory _message = "urgentEnableContractChances";
	 	for(uint i = 0; i < 18; i++){
	 		if (i == 9){
	 			_transactionType = 8;
	 			_message = "urgentDisableTransactionLockChances";
	 		}
	 		__transactions.push(Transaction({
                 transactionID : i,
                 to : address(0),
                 value : 0,
                 data : '',
	 			 transactionTypeStr : transactionTypeStrs[_transactionType], 
                 transactionType : _transactionType,
                 approveCount : 0,
                 canExecuteTimeStamp : block.timestamp,
                 executed : false,
	 			 message : _message,
                 transactionSignature : signature
             	})
	 		);
	 	}
    }

	modifier txExists(uint transactionID) {
		require(transactionID < __transactions.length, "transaction does not exist");
		_;
	}
	modifier notExecuted(uint transactionID) {
		require(!__transactions[transactionID].executed, "transaction already executed");
		_;
	}
	modifier notApproved(uint transactionID) {
		require(!isApproved[transactionID][msg.sender], "msg.sender already approved this transaction");
		_;
	}
	modifier approved(uint transactionID) {
		require(isApproved[transactionID][msg.sender], "msg.sender not approve this transaction");
		_;
	}
	modifier verifyTransactionType(uint _transactionType){
		require(_transactionType <= 11, "Invalid transaction type.");
		_;
	}
	modifier reentrancyLock{
		require(!reentrancyLocked, "No Reentrancy Attack");
		reentrancyLocked = true;
		_;
		reentrancyLocked = false;
	}

    function _getIsApproved(uint transactionID,address owner) public view returns (bool) {
		return isApproved[transactionID][owner];
	}
	function _getSignature() public view returns (string memory) {
		return signature;
	}
    function _getTransactionTypeStrs() public view returns (string[12] memory) {
		return transactionTypeStrs;
	}
	function getNewTransactionID() internal  view returns (uint LastTransactionID) {
		return __transactions.length;
	}
	function _getLastTransactionID() public view returns (uint) {
		return __transactions.length - 1;
	}
}

// File: contracts/DestinyMultiSig.sol
pragma solidity ^0.8.10;
/// @notice Main,destiny multisig contract.
contract IDestinyMultiSig is ITweet,IDIP,IMultiSigTransaction,IMultiSigAdmin,IDestinyDeployer{
	
    modifier verifyApproveCount(uint transactionID) {
		require(
			__transactions[transactionID].approveCount >= minApproveCount,
			"Not enough approvals."
		);
		_;
	}
	modifier verifyExecuteTimeStamp(uint transactionID){	
		require(getTimeStamp() >= __transactions[transactionID].canExecuteTimeStamp,"Execution time not yet reached.");
		_;
	}

	function getTimeStamp() public view returns (uint timeStamp){
		return block.timestamp;
	}
	function getCanExecuteTimeStamp() private view returns (uint timeStamp){
		return getTimeStamp() + transactionLockTime;
	}
	/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/
	/*
	*@Application layer.
    *@User Submit Transaction functions>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>
    */
	/*
	*Issue __transactions, including general __transactions and special __transactions.
	*And automatically approve this transaction for msg.sender.
	*
	*   __transactions.push(
	*       Transaction({
	*           transactionType: _transactionType
	*       })
	*   );
	*
	*   _ApproveTransaction(transactionID);
	*
	*/
    /**
    * @notice Issue __transactions, including general __transactions and manage __transactions.
	*         And automatically approve this transaction for msg.sender.
    *         Only owners callable.
    *         If you need to call other contracts,
    *         please enter the address of the target contract in the (_to) field,
    *         and enter the function signature of the function to be called and the input parameters (if any) in the (_data) field;
    */
    function _IssueTransaction
    (
        address _to,
		uint _value,
		bytes memory _data,
		uint  _transactionType,
		string memory _message
    )
    public
    onlyOwners
    returns (uint transactionID)
    {
        uint __canExecuteTimeStamp = getCanExecuteTimeStamp();
        //It takes an additional 9 days to execute the contract self-destruct.
        if(_transactionType == 11){
            __canExecuteTimeStamp += 777600;

            _DIPs.push(
                DIP({
                    DIP_ID: _DIPs.length,
                    data: string(STORY_END)
                })
            );
            IssueTweet(string(WORLD_LINE_END));
        }
        //DisableTransactionLock and EnableContract __transactions cannot be initiated manually.
        require(_transactionType != 8 && _transactionType !=9,"DisableTransactionLock and EnableContract __transactions cannot be initiated manually.");
        transactionID = __IssueTransaction(_to,_value,_data,_transactionType,_message,__canExecuteTimeStamp);   
        return transactionID;
    }

	function __IssueTransaction
	(
		address _to,
		uint _value,
		bytes memory _data,
		uint  _transactionType,
		string memory _message,
        uint _canExecuteTimeStamp
	)
	internal  
	onlyOwners
	verifyTransactionType(_transactionType)
	verifyContractStauts(_transactionType)
	returns (uint transactionID)
	{
		__transactions.push(
			Transaction({
                transactionID : getNewTransactionID(),
                to : _to,
                value : _value,
                data : _data,
				transactionTypeStr : transactionTypeStrs[_transactionType], 
                transactionType : _transactionType,
                approveCount : 0,
                canExecuteTimeStamp : _canExecuteTimeStamp,
                executed : false,
				message : _message,
                transactionSignature : signature
            })
		);
		emit IssueTransaction(msg.sender, _getLastTransactionID(), _to, __transactions[_getLastTransactionID()]);
		_ApproveTransaction(_getLastTransactionID());
		return _getLastTransactionID();
	}
	/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/
	/*
	*@Approve layer
    *@Approve and Revoke Approve Transaction functions>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>
    */
	/*
	*Approve (transactionID) transaction for msg.sender,
	*
	*   transaction.approveCount += 1;
	*
	*   isApproved[transactionID][msg.sender] = true;
	*
	*And
	*
	*   if(__transactions[transactionID].approveCount >= minApproveCount){
	*        executeTransaction(transactionID);
	*    }
	*
	*/
    /// @notice Approve the transaction corresponding to [transactionID],only owners callable.
	function _ApproveTransaction(uint transactionID)
	public
	onlyOwners
	verifyContractStauts(__transactions[transactionID].transactionType)
	txExists(transactionID)
	notExecuted(transactionID)
	notApproved(transactionID)
	returns (bool transactionapproved)
	{
		Transaction storage transaction = __transactions[transactionID];
		isApproved[transactionID][msg.sender] = true;
		transaction.approveCount += 1;
		emit ApproveTransaction(msg.sender, transactionID);
		if (transaction.approveCount >= minApproveCount && getTimeStamp() >= transaction.canExecuteTimeStamp) {
			_ExecuteTransaction(transactionID);
		}
		return true;
	}

	/*
	*RevokeApprove transaction (transactionID) for msg.sender
	*
	*   transaction.approveCount -= 1;
	*
	*   isApproved[transactionID][msg.sender] = false;
	*
	*/
    /// @notice Unapprove the transaction corresponding to [transactionID] for the caller of the message,only owners callable.
	function _RevokeApprove(uint transactionID)
	public
	onlyOwners
	verifyContractStauts(__transactions[transactionID].transactionType)
	txExists(transactionID)
	notExecuted(transactionID)
	approved(transactionID)
	returns (bool transctionRevokeapproved)
	{
		Transaction storage transaction = __transactions[transactionID];
		isApproved[transactionID][msg.sender] = false;
		transaction.approveCount -= 1;
		emit RevokeApproveaction(msg.sender, transactionID);
		return true;
	}
	/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/
    /*
	*@Executive layer
    *@Transaction execute functions>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>
    */
    /// @notice Execute the transaction corresponding to [transactionID],only owners callable.
	function _ExecuteTransaction(uint transactionID)
	public
	onlyOwners
	verifyContractStauts(__transactions[transactionID].transactionType)
	txExists(transactionID)
	notExecuted(transactionID)
	verifyApproveCount(transactionID)
	verifyExecuteTimeStamp(transactionID)
	reentrancyLock
	returns (bool transactionExecuted)
	{
		Transaction storage transaction = __transactions[transactionID];
		transaction.executed = true;
		/*
		*Determine the transaction type and use the corresponding method to execute the transaction
		*   if(transaction.transactionType == uint)
		*/
		bool success;
		if (transaction.transactionType == 0) {
			success = _executeTransaction(transactionID);
		}
		else if (transaction.transactionType == 1) {
			success = executeIssueDIP(transactionID);
		}
		else if (transaction.transactionType == 2) {
			executeContractDeploy(transactionID);
		}
		else if (transaction.transactionType == 3) {
			success = executeAddOwner(transactionID);
		}
		else if (transaction.transactionType == 4) {
			success = executeRemoveOwner(transactionID);
		}
		else if (transaction.transactionType == 5) {
			success = executeModifySignature(transactionID);
		}
		else if (transaction.transactionType == 6) {
			success = executeModifyMinApproveCount(transactionID);
		}
		else if (transaction.transactionType == 7) {
			success = executeModifyTransactionLock(transactionID);
		}
		else if (transaction.transactionType == 8) {
			success = executeDisableTransactionLock();
		}
		else if (transaction.transactionType == 9) {
			success = executeEnableContract();
		}
		else if (transaction.transactionType == 10) {
			success = executeDisableContract(transactionID);
		}
		else if (transaction.transactionType == 11) {
			executeSelfDestruct(transactionID);
		}
		else {
			require(false, "Invalid TransactionType.");
		}
        require(success,"ExecuteTransaction Fail.");
		emit ExecuteTransaction(msg.sender, transactionID);
		return success;
	}
	/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>The above verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< */

    /// @dev Private execute method.Allows all __transactions to be performed externally using low-level call.Transaction requires (minApproveCount) owner approval.
	function _executeTransaction(uint transactionID)
	private
	onlyOwners
	returns (bool)
	{
		Transaction storage transaction = __transactions[transactionID];
		(bool success,) = transaction.to.call{value : transaction.value}(
			transaction.data
		);
		return success;
	}

	/// @dev Private execute method.Allow IssueDIP.Transaction requires (minApproveCount) owner approval.
	function executeIssueDIP(uint transactionID)
	private
	onlyOwners
	returns (bool)
	{
		_DIPs.push(
			DIP({
				DIP_ID : _DIPs.length,
				data : __transactions[transactionID].message
			})
		);
		emit DIPIssued(_DIPs.length);
		return true;
	}

	/// @dev Private execute method.Allow ContractDeploy.Transaction requires (minApproveCount) owner approval.
	function executeContractDeploy(uint transactionID)
	private
	onlyOwners
	returns (bool)
	{
		// bytes memory _bytecode = __transactions[transactionID].data;
		// bytes32 _salt = keccak256(abi.encodePacked((__transactions[transactionID].message)));
		return deploy(__transactions[transactionID].data,keccak256(abi.encodePacked((__transactions[transactionID].message))));
	}

	/// @dev Private execute method.Allow AddOwner.Transaction requires (minApproveCount) owner approval.
	function executeAddOwner(uint transactionID)
	private
	onlyOwners
	ownerDoesNotExist(__transactions[transactionID].to)
	verifyOwner(__transactions[transactionID].to)
	verifyMinApproveCount(_getMinApproveCount(),owners.length + 1)
	returns (bool)
	{
		address owner = __transactions[transactionID].to;
        isOwner[owner] = true;
		owners.push(owner);
		emit OwnerAddition(owner);
		return true;
	}
	
	/// @dev Private execute method.Allow RemoveOwner.Transaction requires (minApproveCount) owner approval.
	function executeRemoveOwner(uint transactionID)
	private
	onlyOwners
	ownerExists(__transactions[transactionID].to)
	verifyMinApproveCount(_getMinApproveCount(),owners.length - 1)
	returns (bool)
	{
		address owner = __transactions[transactionID].to;
		isOwner[owner] = false;
		if (owners[owners.length - 1] == owner) {
			owners.pop();
		} else {
			for (uint i = 0; i < owners.length - 1; i++) {
				if (owners[i] == owner) {
					owners[i] = owners[owners.length - 1];
					owners.pop();
					break;
				}
			}
		}
		emit OwnerRemoval(owner);
		if (minApproveCount > owners.length) {
			executeModifyMinApproveCount(owners.length);
		}
		return true;
	}
	
	/// @dev Private execute method.Allow ModifySignature.Transaction requires (minApproveCount) owner approval.
	function executeModifySignature(uint transactionID)
	private
	onlyOwners
	returns (bool)
	{
		signature = __transactions[transactionID].message;
		emit SignatureModify(signature);
		return true;
	}
	
	/// @dev Private execute method.Allow ModifyMinApproveCount.Transaction requires (minApproveCount) owner approval.
	function executeModifyMinApproveCount(uint transactionID)
	private
	onlyOwners
	verifyMinApproveCount(__transactions[transactionID].value,owners.length)
	returns (bool)
	{
		minApproveCount = __transactions[transactionID].value;
		emit MinApproveCountModify(minApproveCount);
		return true;
	}

	/// @dev Private execute method.Allow ModifyTransactionLock.Transaction requires (minApproveCount) owner approval.
	function executeModifyTransactionLock(uint transactionID)
	private
	onlyOwners
	verifyTransactionLockTime(__transactions[transactionID].value)
	returns (bool)
	{
		transactionLockTime = __transactions[transactionID].value;
		emit TransactionLockModify(transactionLockTime);
		return true;
	}

	/// @dev Private execute method.Allow disable TransactionLock.Transaction requires (minApproveCount) owner approval.
	function executeDisableTransactionLock()
	private 
	onlyOwners
	returns (bool)
	{
		transactionLockTime = 0;
		emit TransactionLockModify(transactionLockTime);
		return true;
	}

	/// @dev Private execute method.Allow enable the contract.Transaction requires (minApproveCount) owner approval.
	function executeEnableContract()
	private
	onlyOwners
	returns (bool)
	{
		contractDisabled = false;
		emit ContractEnabled(true, getTimeStamp());
		return true;
	}

	/// @dev Private execute method.Allows disabling all functionality of the contract except for deposits and _tweets.Transaction requires (minApproveCount) owner approval.
	function executeDisableContract(uint transactionID)
	private
	onlyOwners
	returns (bool)
	{
		__IssueTransaction(address(this),0,'',9,"This transaction is automatically initiated by function executeDisableContract.",getTimeStamp()+__transactions[transactionID].value);		
		contractDisabled = true;
        emit ContractDisabled(getTimeStamp() + __transactions[transactionID].value);
        return true;
	}

	/// @dev Private execute method.Allows to contract execute selfdestruct.Transaction requires (minApproveCount) owner approval.
	function executeSelfDestruct(uint transactionID)
	private
	onlyOwners
	{
		require(__transactions[transactionID].to != address(0), "UltimateBeneficiaryAddress Can't is zero address.");
        address _ultimateBeneficiaryAddress = __transactions[transactionID].to;
        emit SelfDestruction(_ultimateBeneficiaryAddress);
        selfdestruct(payable(_ultimateBeneficiaryAddress));
	}
}

// File: contracts/DestinyTemple_Token.sol
pragma solidity ^0.8.10;
/// @notice Main,Destiny Governance Token.
contract IDestinyTempleToken is IERC20{
    address constant private OWNER = address(0);
    string constant private NAME = "DestinyTemple";
    string constant private SYMBOL = "DST";
    uint constant private DECIMALS = 0;
    uint256 constant private TOTALSUPPLY = 19999;
    uint256 constant private MAX_UINT256 = 2**256 - 1;
    mapping (address => uint256) private  _balances;
    mapping (address => mapping (address => uint256)) private  _allowances;

    modifier msgSenderNotZero(){
        require(msg.sender != address(0), "ERC20: transfer from the zero address.");
        _;
    }
    modifier verifyBalance(address sender,uint256 value){
        require(_balances[sender] >= value,"ERC20: transfer amount exceeds balance.");
        _;
    }
    modifier verifyAllowance(uint256 _allowance,uint256 value){
        require(_allowance >= value,"ERC20: transfer amount exceeds allowance");
        _;
    }
    constructor(){            
        _balances[0x00001C1D6ab92F943eD4A31dA8F447Fd96589960] = TOTALSUPPLY;
    }
    function owner() external pure returns (address) {
        return OWNER;
    }
    function name() external pure returns (string memory) {
        return NAME;
    }
    function symbol() external pure returns (string memory) {
        return SYMBOL;
    }
    function decimals() external pure returns (uint) {
        return DECIMALS;
    }
    function totalSupply() external pure returns (uint256) {
        return TOTALSUPPLY;
    }
    function balanceOf(address _owner) external view returns (uint256) {
        return _balances[_owner];
    }
    function allowance(address _owner, address _spender) external view returns (uint256 remaining) {
        return _allowances[_owner][_spender];
    }
    /// @notice Transfer _value amount of tokens to recipient;
    function transfer(address recipient, uint256 _value) external msgSenderNotZero verifyBalance(msg.sender,_value)  returns (bool) {
        _balances[msg.sender] -= _value;
        _balances[recipient] += _value;

        emit Transfer(msg.sender, recipient, _value);
        return true;
    }
    ///@notice Transfer _value tokens from the sender who has authorized you to the recipient.
    function transferFrom(address sender, address recipient, uint256 _value) external msgSenderNotZero verifyBalance(sender,_value) verifyAllowance(_allowances[sender][msg.sender],_value) returns (bool) {        
        _balances[sender] -= _value;
        _balances[recipient] += _value;
        if (_allowances[sender][msg.sender] < MAX_UINT256) {
            _allowances[sender][msg.sender] -= _value;
        }

        emit Transfer(sender, recipient, _value);
        return true;
    }
    /// @notice Grant _spender the right to control your _value amount of the token.
    function approve(address sender, uint256 _value) external  msgSenderNotZero returns (bool success) {
        _allowances[msg.sender][sender] = _value;

        emit Approval(msg.sender, sender, _value);
        return true;
    }
}

// File: DestinyMultiSig.sol
pragma solidity ^0.8.10;
/**
 * @notice The final contract, which integrates all of the above.
 *         The contract can accept ETH, ERC20, ERC721, ERC1155 token deposits.
 */
contract DestinyTemple is IDestinyMultiSig,IDestinyTempleToken,IProxyAdmin,IERC721Receiver,IERC1155Receiver{
    /// @notice DestinyTemple_Initialize onlyDESTINY_EXCUTORS;
    function DestinyTemple_Initialize() external verifyInitialize onlyDESTINY_EXCUTORS returns(bool){
        IProxyAdmin_Initialize();
        return true;
    }
    fallback()external payable {
		emit Deposit(msg.sender, msg.value, address(this).balance);
	}
	receive() external payable {
		emit Deposit(msg.sender, msg.value, address(this).balance);
	}

    /// See ERC 165.
	function supportsInterface(bytes4 interfaceId) external pure  override returns (bool){
		return interfaceId == type(IERC1155Receiver).interfaceId || interfaceId == type(IERC165).interfaceId ||interfaceId == type(IERC721Receiver).interfaceId;
	}

    /// @notice Used to receive ERC721(NFT) tokens.
	function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external pure override  returns (bytes4){
		return IERC721Receiver.onERC721Received.selector;
	}

    /// @notice Used to receive ERC1155 tokens.
	function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes calldata data
    ) external pure override returns (bytes4){
		return IERC1155Receiver.onERC1155Received.selector;
	}

    /// @notice Used to receive ERC1155 tokens in batches.
    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] calldata ids,
        uint256[] calldata values,
        bytes calldata data
    ) external pure override returns (bytes4){
		return IERC1155Receiver.onERC1155BatchReceived.selector;
	}
}
/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>Contract verification passed.<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<*/

/*=======================================================================================================================
#            __                                 __                               __                                     #
#           /  \ |  o     _  ._ _  o     _.    /  \  _  _  | _|_  _ |  _.       /  \    o _  o                          #
#          | (|/ |< | \/ (_) | | | | \/ (_|   | (|/ _> (_) |  |_ (_ | (_| \/   | (|/ >< | /_ |                          #
#           \__       /              /         \__                        /     \__                                     #
#            __                                  __                                    __                               #
#           /  \  _             _. o ._   _     /  \  _|                    _. ._     /  \  _ o  _  _.  _|  _.          #
#          | (|/ _> |_| \/ |_| (_| | | | (_|   | (|/ (_| |_| \/ |_| >< |_| (_| | |   | (|/ (_ | (_ (_| (_| (_|          #
#           \__         /        |        _|    \__          /                        \__                               #
#            __                             __                    __              __                                    #
#           /  \ o o  _. ._          _     /  \     ._ _   _     /  \     | o    /  \ |_   _. ._    |   _.              #
#          | (|/ | | (_| | | \/ |_| (/_   | (|/ |_| | | | (_)   | (|/ |_| | |   | (|/ | | (_| | |_| |< (_|              #
#           \__ _|           /             \__                   \__             \__                                    #
#                                                                                                                       #
=========================================================================================================================
#                                                                                                                       #
#     ██████╗ ███████╗███████╗████████╗██╗███╗   ██╗██╗   ██╗████████╗███████╗███╗   ███╗██████╗ ██╗     ███████╗       #
#     ██╔══██╗██╔════╝██╔════╝╚══██╔══╝██║████╗  ██║╚██╗ ██╔╝╚══██╔══╝██╔════╝████╗ ████║██╔══██╗██║     ██╔════╝       #
#     ██║  ██║█████╗  ███████╗   ██║   ██║██╔██╗ ██║ ╚████╔╝    ██║   █████╗  ██╔████╔██║██████╔╝██║     █████╗         #
#     ██║  ██║██╔══╝  ╚════██║   ██║   ██║██║╚██╗██║  ╚██╔╝     ██║   ██╔══╝  ██║╚██╔╝██║██╔═══╝ ██║     ██╔══╝         #
#     ██████╔╝███████╗███████║   ██║   ██║██║ ╚████║   ██║      ██║   ███████╗██║ ╚═╝ ██║██║     ███████╗███████╗       #
#     ╚═════╝ ╚══════╝╚══════╝   ╚═╝   ╚═╝╚═╝  ╚═══╝   ╚═╝      ╚═╝   ╚══════╝╚═╝     ╚═╝╚═╝     ╚══════╝╚══════╝       #
#                                                                                                                       #
========================================================================================================================*/